

5/7.1

Effets vidéo sur CPC

Peut-être avez-vous remarqué dans certains jeux ou utilitaires, des effets vidéo agréables, au demeurant fort simples, mais qui produisent toujours leur effet de surprise.

Nous vous proposons dans ce chapitre de créer quelques instructions sous forme de RSX qui vous permettront d'effectuer des présentations intéressantes sur l'écran de votre micro-ordinateur préféré.

Ces nouvelles instructions, utilisent pour la plupart les registres du composant s'occupant de la gestion de l'écran des CPC, c'est-à-dire le 6845 de chez Motorola. Nous vous conseillons à cette occasion de relire les pages de la Partie 2 Chapitre 3.2 traitant de ce composant.

PROGRAMMATION DU 6845

Après avoir pris connaissance des registres du 6845, vous savez maintenant que pour accéder à l'un des registres de programmation de celui-ci (registres de contrôle R0 à R17), il vous faut donner le numéro de ce registre dans le registre d'adresse **AR**.

Le registre d'adresse **AR** se trouve à l'adresse numéro **&BC00**, le (ou les) registre(s) de contrôle se trouvant à l'adresse **&BD00**.

Ainsi en Basic, pour charger le registre de synchronisation horizontal avec la valeur **&4F**, il faudra entrer :

```
OUT &BC00,2    : REM NUMERO DU REGISTRE R2
OUT &BD00,&4F  : REM MODIFICATION
```

Cela donne, en assembleur :

```
LD    A,2      ; NUMERO DU REGISTRE
LD    BC,0BC00H ; ADRESSE AR
OUT   (C),A    ; ACCES R2
LD    A,04FH   ; VALEUR
LD    BC,0BD00H ; ADRESSE REGISTRE CONTROL
OUT   (C),A    ; MODIFICATION
```

LES NOUVELLES INSTRUCTIONS BASIC

I INIT6845

Comme nous allons modifier les caractéristiques des registres du 6845, nous avons, en premier lieu, pensé à créer une instruction permettant de reconfigurer l'état d'initialisation de ce composant.

Cette instruction vous permettra aussi, lorsque vous voudrez expérimenter quelques effets personnels, de restaurer l'écran qui sera peut-être perdu.

I SCROLLPRINT

Cette instruction dont la syntaxe est la suivante :

I SCROLLPRINT,positionx,positiony,@TEXTE\$

permet d'afficher à la position **positionx,positiony** (comme **LOCATE positionx,positiony**), le message situé dans la variable alphanumérique **TEXTE\$**.

Rien de bien original nous direz-vous ! Si ! Le message s'affichera caractère par caractère, avec un effet de scrolling arrière surprenant.

I ROULEH

Cette instruction fait disparaître le contenu de l'écran de votre CPC avec un effet de rouleau horizontal.

Il est possible de restaurer instantanément le contenu de celui-ci grâce à l'instruction **I INIT6845**.

Une utilisation intéressante consisterait à créer cet écran suite à l'utilisation de **I ROULEH**, c'est-à-dire en mode non visualisation, puis de le faire apparaître instantanément.

I DEROULEH

Cette instruction effectue le mouvement inverse sur l'écran du CPC et restitue le contenu caché de l'écran en le déroulant.

I TREMBLE

Cette instruction provoque le tremblement de l'écran dans le sens vertical.

L'arrêt de l'effet de cette instruction s'effectue par l'utilisation de **I INIT6845**.

I ONDULE

Cette instruction envire votre écran en provoquant une légère ondulation du contenu de celui-ci.

Vous pourrez accentuer ou adoucir celui-ci en modifiant le réglage du bouton de synchronisation de votre moniteur.

Pour annuler cet effet, un seul remède : **I INIT6845**.

I ECRASE

Cette instruction provoquera un effet d'écrasement de toute l'image située sur l'écran. Effet qui pourra faire croire à un défaut sur l'ordinateur d'un de vos amis.

Là encore, on restitue l'état initial de l'écran à l'aide de l'INIT6845.

I DESTROY

Pour aller plus loin dans la dégradation fictive de votre écran, voici une instruction qui modifie complètement la synchronisation de l'écran et qui en surprendra certainement plus d'un !

Le service après-vente vous conseille d'entrer, en mode aveugle, ou à tâtons, l'instruction l'INIT6845.

LE PROGRAMME ASSEMBLEUR

L'écriture de ce programme créant de nouvelles instructions passe automatiquement par l'assembleur, mais, rassurons tout de suite les programmeurs férus du Basic, nous leur fournirons le chargeur adéquat.

Le listing assembleur est le suivant :

```

1          ;*****
2          ;*          *
3          ;*  INSTRUCTIONS RSXs *
4          ;*    PROPOSANT DES   *
5          ;*    EFFETS  VIDEO  *
6          ;*          *
7          ;*****
8          ;
9          ;
10         ;*** TABLE DES EQUIVALENCES ***
11         ;
12         ;INSERTION DES RSX
13         ;    KL-LOG-TEXT
14         KLOGTX:    EQU  0BCD1H
15         ;
16         ;REGISTRE D'ADRESSE 6845
17         ;    ADRESS REGISTER
18         AREG:     EQU  0BC00H
19         ;
20         ;REGISTRES DE CONTROLE 6845
21         ;    CONTROL REGISTER
22         CREG:     EQU  0BD00H
23         ;
24         ;VECTEUR POUR MODE ECRAN
25         ;    SCR-GET-MODE
26         SGMODE:   EQU  0BC11H
27         ;
28         ;POSITIONNEMENT CURSEUR
29         ;    TXT-SET-CURSOR
30         TSCURS:   EQU  0BB75H
31         ;
32         ;AFFICHAGE CARACTERE
33         ;    TXT-OUTPUT
34         TXTOUT:   EQU  0BB5AH
35         ;
36         ;
37         ;*****
38         ;
39         ;
40         ;*****
41         ;*  EMPLACEMENT DU   *
42         ;*    PROGRAMME     *
43         ;*****
44         ;
45         ;          ORG  0A000H
46         ;
47         ;          LOAD 0A000H
48         ;
49         ;
50         ;*****
51         ;*  INSERTION DES RSXs *
52         ;*****
53         ;
54         DEBUT:    EQU  $
55         A000 210FA0    LD  HL,KERNEL          ;ARESERVER POUR RSX
56         A003 0113A0    LD  BC,VECTEU         ;TABLE INSTRUCTIONS
57
58         A006 3EC9      LD  A,0C9H           ;POUR EVITER UN
59         A008 3200A0    LD  (DEBUT),A        ;NOUVEL APPEL
60

```

```

61 A00B CDD1BC          CALL KLOGTX          ;INITIALISE RSXs
62
63 A00E C9              RET                ;FIN
64
65                      ;
66                      ;*** RESERVE POUR KERNEL ***
67                      ;
68                      KERNEL:      EQU $
69 A00F 00000000        DEFB 00,00,00,00
70                      ;
71                      ;*** TABLE INSTRUCTIONS ***
72                      ;
73                      VECTEU:      EQU $
74 A013 2DA0           DEFW TABLE
75                      ;
76                      ;*** TABLE DES SAUTS ***
77                      ;
78 A015 C368A0         JP    IN6845
79 A018 C390A0         JP    ROULE
80 A01B C3ABA0         JP    DEROUL
81 A01E C3C7A0         JP    TREMBL
82 A021 C3D6A0         JP    ECRASE
83 A024 C3E5A0         JP    SCRTXT
84 A027 C384A1         JP    DESTRO
85 A02A C393A1         JP    ONDULE
86                      ;
87                      ;
88                      ;*** SYNTAXE INSTRUCTIONS ***
89                      ;
90                      TABLE:      EQU $
91                      ;
92 A02D 494E4954        DEFB "INIT684"
92 A031 363834
93 A034 B5             DEFB "5"+080H
94                      ;
95 A035 524F554C        DEFB "ROULE"
95 A039 45
96 A03A C8             DEFB "H"+080H
97                      ;
98 A03B 4445524F        DEFB "DEROULE"
98 A03F 554C45
99 A042 C8             DEFB "H"+080H
100                      ;
101 A043 5452454D        DEFB "TREMBL"
101 A047 424C
102 A049 C5             DEFB "E"+080H
103                      ;
104 A04A 45435241        DEFB "ECRAS"
104 A04E 53
105 A04F C5             DEFB "E"+080H
106                      ;
107 A050 5343524F        DEFB "SCROLL"
107 A054 4C4C
108 A056 5052494E        DEFB "PRIN"
109 A05A D4             DEFB "T"+080H
110                      ;
111 A05B 44455354        DEFB "DESTRO"
111 A05F 524F
112 A061 D9             DEFB "Y"+080H
113                      ;

```

```

114 A062 4F4E4455          DEFB "ONDUL"
114 A066 4C
115 A067 C5              DEFB "E"+080H
116                      ;
117                      ;
118                      ;*****
119                      ;*                *
120                      ;*  EXECUTION DES  *
121                      ;*  DIFFERENTES   *
122                      ;*  RSXs VIDEO    *
123                      ;*                *
124                      ;*****
125                      ;
126                      ;
127                      ;*** INITIALISATION 6845 ***
128                      ;
129                      IN6845:      EQU $
130 A068 2190A0          LD  HL,FININI          ;POINTE FIN TABLE
131                      ;D'INITIALISATION DU 6845
132                      ;POUR UNE FREQUENCE DE 50 Hz
133 A06B 0100BC          LD  BC,AREG          ;REGISTRE D'ADRESSE
134 A06E 0E0F           LD  C,OFH           ;REGISTRE 15
135 A070 2B             DEC  HL
136                      ;
137                      I68452:     EQU $
138 A071 ED49           OUT  (C),C          ;POINTE REGISTRE CONTROL
139 A073 04            INC  B           ;REGISTRE DE CONTROL
140 A074 7E           LD  A,(HL)         ;VALEUR DE LA TABLE
141 A075 ED79           OUT  (C),A          ;DANS LE REGISTRE CONTROL
142 A077 05           DEC  B           ;REGISTRE D'ADRESSE
143 A078 0D           DEC  C           ;SUIVANT OU PRECEDENT
144 A079 2B           DEC  HL
145                      ;LA TABLE
146 A07A 79           LD  A,C           ;REG CONTROL DANS A
147 A07B FEFF          CP  OFFH          ;SI TOUS Y SONT PASSES
148 A07D 20F2          JR  NZ,I68452      ;SINON ENCORE
149 A07F C9           RET
150                      ;
151                      I68451:     EQU $          ;TABLE 50 Hz
152 A080 3F           DEFB 03FH          ;H-TOTAL
153 A081 28           DEFB 028H          ;H-DISPLAY
154 A082 2E           DEFB 02EH          ;H-SYNC
155 A083 8E           DEFB 08EH          ;SYNC-WIDTH
156 A084 26           DEFB 026H          ;V-TOTAL
157 A085 00           DEFB 00H          ;V-ADJUST
158 A086 19           DEFB 019H          ;V-DISPLAY
159 A087 1E           DEFB 01EH          ;V-SYNC
160 A088 00           DEFB 00H          ;INTERLACE
161 A089 07           DEFB 07H          ;SCAN LINE
162 A08A 00           DEFB 00H          ;CURSOR START
163 A08B 00           DEFB 00H          ;CURSOR END
164 A08C 3000          DEFB 030H,00H        ;START RAM
165 A08E C000          DEFB 0C0H,00H        ;CURSOR POS.
166                      FININI:     EQU $
167                      ;
168                      ;
169                      ;*** EFFACEMENT ECRAN PAR ***
170                      ;***   MODE ROULEAU   ***
171                      ;***   HORIZONTAL   ***
172                      ;

```

```

173          ROULE:      EQU  $
174 A090 3E27          LD  A,027H          ;40 CARACTERES DEFINIS
175          ;
176          ROUL3:     EQU  $
177 A092 0100BC        LD  BC,AREG          ;POINT ADRESS REGISTER
178 A095 0E01          LD  C,01           ;REGISTRE 1
179 A097 ED49          OUT (C),C
180 A099 04            INC  B              ;POINTE CONTROL REGISTER
181 A09A ED79          OUT (C),A          ;EFFET REUSSI
182 A09C 47            LD  B,A              ;SAUVE A
183          ;
184 A09D 210020        LD  HL,02000H        ;TEMPO
185          ROUL4:     EQU  $
186 A0A0 2B            DEC  HL
187 A0A1 7C            LD  A,H
188 A0A2 B5            OR   L
189 A0A3 20FB          JR   NZ,ROUL4
190 A0A5 78            LD  A,B              ;RECUPERE A
191 A0A6 3D            DEC  A              ;1 CARACTERE DE MOINS
192 A0A7 04            INC  B
193 A0A8 10E8          DJNZ ROUL3          ;AU SUIVANT
194          ;
195 A0AA C9            RET              ;FIN
196          ;
197          ;
198          ;*** REMISE EN FORME HORIZONTALE *
199          ;*** DE L'ECRAN VIDEO *
200          ;
201          DEROUL:    EQU  $
202 A0AB 3E00          LD  A,00H          ;ON COMMENCE A ZERO
203          DEROU3:    EQU  $
204 A0AD 0100BC        LD  BC,AREG
205 A0B0 0E01          LD  C,01H          ;REGISTRE 1
206 A0B2 ED49          OUT (C),C          ;EN ACCES
207 A0B4 04            INC  B              ;CONTROL REGISTER
208 A0B5 ED79          OUT (C),A          ;EFFET VIDEO
209 A0B7 F5            PUSH AF            ;SAUVE AF
210          ;
211 A0B8 210020        LD  HL,02000H        ;TEMPO
212          DEROU4:    EQU  $
213 A0BB 2B            DEC  HL
214 A0BC 7C            LD  A,H
215 A0BD B5            OR   L
216 A0BE 20FB          JR   NZ,DEROU4
217          ;
218 A0C0 F1            POP  AF              ;RECUPERE AF
219 A0C1 3C            INC  A              ;UN CARACTERE DE PLUS
220 A0C2 FE29          CP   029H          ;PLUS DE 40?
221 A0C4 20E7          JR   NZ,DEROU3     ;NON, RECOMMENCER
222          ;
223 A0C6 C9            RET              ;OUI, FIN
224          ;
225          ;
226          ;*** EFFET DE TREMBLEMENT ***
227          ;
228          TREMBL:    EQU  $
229 A0C7 0100BC        LD  BC,AREG          ;ADRESS REGISTER
230 A0CA 3E08          LD  A,08H          ;REGISTRE 8
231 A0CC ED79          OUT (C),A          ;EN ACCES
232 A0CE 0100BD        LD  BC,CREG          ;CONTROL REGISTER

```

```

233 A0D1 3E01          LD  A,01H          ;MODE ENTRELACE
234 A0D3 ED79          OUT (C),A          ;EFFET
235                    ;
236 A0D5 C9            RET                    ;FIN
237                    ;
238                    ;
239                    ;*** EFFET ECRASEMENT ***
240                    ;
241 ECRASE:             EQU  $
242 A0D6 0100BC         LD  BC,AREG          ;ADRESS REGISTER
243 A0D9 3E08           LD  A,08H           ;REGISTRE 8
244 A0DB ED79           OUT (C),A          ;EN ACCES
245 A0DD 0100BD         LD  BC,CREG          ;CONTROL REGISTER
246 A0E0 3E03           LD  A,03H           ;MODE ENTRELACE VIDEO
247 A0E2 ED79           OUT (C),A          ;EFFET
248                    ;
249 A0E4 C9            RET                    ;FIN
250                    ;
251                    ;
252                    ;*** SROLLING HORIZONTAL ***
253                    ;*** POUR L'AFFICHAGE ***
254                    ;*** D'UN TEXTE ***
255                    ;
256 SCRTXT:             EQU  $
257 A0E5 FE03           CP   03H            ;3 PARAMETRES?
258 A0E7 205A           JR   NZ,ERREU1     ;NON ERREUR
259                    ;
260 A0E9 DD7E04         LD  A,(IX+4)        ;RECUPERE X
261 A0EC FE50           CP   050H           ;PLUS QUE 80?
262 A0EE D268A1         JP  NC,ERREU2      ;OUI ERREUR
263 A0F1 DD7E02         LD  A,(IX+02H)     ;RECUPERE Y
264 A0F4 FE19           CP   019H           ;PLUS QUE 25?
265 A0F6 D268A1         JP  NC,ERREU2      ;OUI ERREUR
266                    ;
267 A0F9 DD6601         LD  H,(IX+01H)     ;ADRESSE HAUTE
268 A0FC DD6E00         LD  L,(IX+00H)     ;ADRESSE BASSE
269 A0FF 7E             LD  A,(HL)         ;NB DE LETTRES
270 A100 3240A1        LD  (NBLET),A      ;A RANGER
271 A103 23            INC  HL            ;SUIVANT
272 A104 7E             LD  A,(HL)         ;ADRESSE BASSE
273 A105 3242A1        LD  (OCTET0),A     ;CHAINE A RANGER
274 A108 23            INC  HL            ;SUIVANT
275 A109 7E             LD  A,(HL)         ;ADRESSE HAUTE
276 A10A 3241A1        LD  (OCTET1),A     ;CHAINE A RANGER
277                    ;
278 A10D 0601          LD  B,01H          ;PREMIER CARACTERE
279                    ;
280 SCRTX2:             EQU  $
281 A10F C5             PUSH BC            ;SAUVER
282 A110 DD7E04         LD  A,(IX+04H)     ;REPREND X
283 A113 C1             POP  BC            ;RECUPERE BC
284 A114 C5             PUSH BC            ;RESAUVE
285 A115 90            SUB  B             ;X-I
286 A116 E5            PUSH HL           ;SAUVE HL
287 A117 2140A1        LD  HL,NBLET      ;ADRESSE
288 A11A 86            ADD  A,(HL)        ;X-I+LEN(CHAINES)
289 A11B E1            POP  HL           ;RECUPERE HL
290 A11C 67            LD  H,A           ;POSITIONX DANS A
291 A11D DD6E02         LD  L,(IX+02H)     ;PREND Y
292 A120 CD75BB        CALL TSCURS       ;POSITION CURSEUR

```



```

293                                     ;
294 A123 C1                             POP BC                               ;REPREND BC
295 A124 C5                             PUSH BC                              ;RESAUVER
296 A125 3A41A1                         LD A,(OCTET1)                       ;ADRESSE HAUTE
297 A128 67                             LD H,A                              ;CHAINE DANS H
298 A129 3A42A1                         LD A,(OCTET0)                       ;ADRESSE BASSE
299 A12C 6F                             LD L,A                              ;DE CHAINE DANS L
300                                     ;
301 SCRTX1: EQU $
302 A12D 7E                             LD A,(HL)                           ;LETTE EN POSITION HL
303 A12E E5                             PUSH HL                              ;A SAUVER
304 A12F CD5ABB                         CALL TXTOUT                          ;AFFICHER CARACTERE
305                                     ;
306 A132 E1                             POP HL                               ;RECUPERE HL
307 A133 23                             INC HL                               ;SUIVANT
308 A134 10F7                         DJNZ SCRTX1                          ;ENCORE UN CARACTERE?
309                                     ;
310 A136 C1                             POP BC                               ;REPREND BC
311 A137 04                             INC B                               ;DECALE POSITION
312 A138 3A40A1                         LD A,(NBLET)                        ;RECUPERE LONGUEUR
313 A13B 3C                             INC A                               ;PLUS UNE
314 A13C B8                             CP B                                 ;MESSAGE ENTIER?
315 A13D 20D0                         JR NZ,SCRTX2                        ;NON RECOMMENCER
316                                     ;
317 A13F C9                             RET                                  ;FIN
318                                     ;
319 NBLET: EQU $
320 A140 00                             DEFB 00H                            ;RESERVE POUR LONGUEUR
321                                     ;
322 OCTET1: EQU $
323 A141 00                             DEFB 00H
324 OCTET0: EQU $
325 A142 00                             DEFB 00H
326                                     ;
327                                     ;
328 ERREU1: EQU $
329 A143 214FA1                         LD HL,TEXTE1                        ;1ER MESSAGE
330                                     ;
331 ERR11: EQU $
332 A146 7E                             LD A,(HL)                           ;AFFICHAGE
333 A147 B7                             OR A                                 ;DU
334 A148 C8                             RET Z                               ;MESSAGE
335 A149 CD5ABB                         CALL TXTOUT                          ;POINTE
336 A14C 23                             INC HL                              ;PAR
337 A14D 18F7                         JR ERR11                            ;REGISTRE HL
338                                     ;
339 TEXTE1: EQU $
340 A14F 0A0D                         DEFB 0AH,ODH
341 A151 494C2046                     DEFB "IL FAUT "
341 A155 41555420
342 A159 33205041                     DEFB "3 PARAMETRES"
342 A15D 52414D45
342 A161 54524553
343 A165 0A0D                         DEFB 0AH,ODH
344 A167 00                             DEFB 00H
345                                     ;
346 ERREU2: EQU $
347 A168 216DA1                         LD HL,TEXTE2                        ;2EE MESSAGE
348 A16B 18D9                         JR ERR11                            ;A AFFICHER
349                                     ;

```

```

350          TEXTE2:      EQU  $
351 A16D 0A0D          DEFB 0AH,ODH
352 A16F 45525245     DEFB "ERREUR DE "
352 A173 55522044
352 A177 4520
353 A179 504F5349     DEFB "POSITION"
353 A17D 54494F4E
354 A181 0A0D          DEFB 0AH,ODH
355 A183 00           DEFB 00H
356          ;
357          ;
358          ;*** EFFET DE DESTRUCTION ***
359          ;***     SUR L'ECRAN     ***
360          ;***     DU CPC     ***
361          ;
362          DESTRO:      EQU  $
363 A184 0100BC        LD  BC,AREG          ;EDRESS REGISTER
364 A187 0E09          LD  C,09H           ;MAX RASTER ADRESS
365 A189 ED49          OUT (C),C
366 A18B 0100BD        LD  BC,CREG
367 A18E 3E00          LD  A,00H           ;VALEUR 0
368 A190 ED79          OUT (C),A
369          ;
370 A192 C9            RET              ;FIN
371          ;
372          ;
373          ;*** EFFET D'ONDULATION ***
374          ;
375          ONDULE:      EQU  $
376 A199 0100BC        LD  BC,AREG          ;ADRESS REGISTER
377 A196 3E03          LD  A,03H
378 A198 ED79          OUT (C),A
379 A19A 0100BD        LD  BC,CREG          ;CONTROL REGISTER
380 A19D 3E01          LD  A,01H
381 A19F ED79          OUT (C),A
382          ;
383 A1A1 C9            RET              ;FIN
384          ;
385          ;
386          END

```

Lignes 9 à 34 : définitions des équivalences utilisées

Lignes 54 à 115 : définition des instructions

Lignes 129 à 166 : instruction I INIT6845

Lignes 173 à 195 : instruction I ROULEH

Lignes 201 à 223 : instruction I DEROULEH

Lignes 229 à 236 : instruction I TREMBLE

Lignes 242 à 249 : instruction I ECRASE

Lignes 257 à 355 : instruction I SCROLLPRINT

Lignes 363 à 370 : instruction I DESTROY

Lignes 376 à 383 : instruction I ONDULE

LE CHARGEUR BASIC

```

10 REM *****
20 REM **
30 REM ** GENERATION DES CODES **
40 REM ** MACHINES POUR CREER **
50 REM ** DES RSXs D'EFFETS **
60 REM ** VIDEO **
70 REM **
80 REM *****
90 REM
100 REM *** RESERVATION MEMOIRE ***
110 REM
120 MEMORY &9FFF
130 OPENOUT "MACHIN."
140 MEMORY HIMEM-1
150 CLOSEOUT
160 REM
170 controle = 0
180 REM
190 REM *** MISE EN MEMOIRE ***
200 REM
210 FOR A = &A000 TO &A1A1
220 READ B$:POKE A,VAL("&" + B$)
230 controle = controle + VAL("&" + B$)
240 NEXT A
250 REM
260 IF controle = 77713 THEN GOTO 380
270 MODE 2
280 PRINT CHR$(7)
290 PRINT "Vous avez fait une erreur"
300 PRINT "en entrant les DATAs ou le nombre de controle 77713"
310 PRINT "Verifiez les svp"
320 REM
330 PRINT:PRINT
340 STOP
350 REM
360 REM *** sauvegarde ***
370 REM
380 MODE 2
390 PRINT "SAUVEGARDE DU PROGRAMME BINAI
RE CREE"
400 PRINT:PRINT
410 SAVE "EFFETVID.BIN",B,&A000,&1A2
420 PRINT:PRINT
430 PRINT "Le programme est sauvegarde sous le nom"
440 PRINT
450 PRINT " EFFETSON.BIN"
460 PRINT:PRINT

```

```
470 PRINT "Pour utiliser le fichier bina
ire ainsi cree,"
480 PRINT "inserez dans votre programme
les lignes suivantes:
490 PRINT "      MEMORY &9FFF"
500 PRINT "      OPENOUT " + CHR$(34) + "
BIDON.BID"+CHR$(34)
510 PRINT "      MEMORY HIMEM - 1"
520 PRINT "      CLOSEOUT"
530 PRINT "      LOAD " + CHR$(34) + "EFF
ETVID.BIN" + CHR$(34) + ",&A000"
540 PRINT "      CALL &A000"
550 PRINT "      ... suite du programme
- -"
560 PRINT:PRINT
570 END
580 REM
590 REM *** CODES HEXADECIMAUX ***
600 REM
610 REM *** BLOC 1 ***
620 REM
630 DATA 21,0F,A0,01,13,A0,3E,C9
640 DATA 32,00,A0,CD,D1,BC,C9,00
650 DATA 00,00,00,2D,A0,C3,68,A0
660 DATA C3,90,A0,C3,AB,A0,C3,C7
670 DATA A0,C3,D6,A0,C3,E5,A0,C3
680 DATA 84,A1,C3,93,A1,49,4E,49
690 DATA 54,36,38,34,B5,52,4F,55
700 DATA 4C,45,C8,44,45,52,4F,55
710 REM
720 REM *** BLOC 2 ***
730 REM
740 DATA 4C,45,C8,54,52,45,4D,42
750 DATA 4C,C5,45,43,52,41,53,C5
760 DATA 53,43,52,4F,4C,4C,50,52
770 DATA 49,4E,D4,44,45,53,54,52
780 DATA 4F,D9,4F,4E,44,55,4C,C5
790 DATA 21,90,A0,01,00,BC,0E,0F
800 DATA 2B,ED,49,04,7E,ED,79,05
810 DATA 0D,2B,79,FE,FF,20,F2,C9
820 REM
830 REM *** BLOC 3 ***
840 REM
850 DATA 3F,28,2E,8E,26,00,19,1E
860 DATA 00,07,00,00,30,00,C0,00
870 DATA 3E,27,01,00,BC,0E,01,ED
880 DATA 49,04,ED,79,47,21,00,20
890 DATA 2B,7C,B5,20,FB,78,3D,04
900 DATA 10,E8,C9,3E,00,01,00,BC
910 DATA 0E,01,ED,49,04,ED,79,F5
```

```
920 DATA 21,00,20,2B,7C,B5,20,FB
930 REM
940 REM *** BLOC 4 ***
950 REM
960 DATA F1,3C,FE,29,20,E7,C9,01
970 DATA 00,BC,3E,08,ED,79,01,00
980 DATA BD,3E,01,ED,79,C9,01,00
990 DATA BC,3E,08,ED,79,01,00,BD
1000 DATA 3E,03,ED,79,C9,FE,03,20
1010 DATA 5A,DD,7E,04,FE,50,D2,68
1020 DATA A1,DD,7E,02,FE,19,D2,68
1030 DATA A1,DD,66,01,DD,6E,00,7E
1040 REM
1050 REM *** BLOC 5 ***
1060 REM
1070 DATA 32,40,A1,23,7E,32,42,A1
1080 DATA 23,7E,32,41,A1,06,01,C5
1090 DATA DD,7E,04,C1,C5,90,E5,21
1100 DATA 40,A1,86,E1,67,DD,6E,02
1110 DATA CD,75,BB,C1,C5,3A,41,A1
1120 DATA 67,3A,42,A1,6F,7E,E5,CD
1130 DATA 5A,BB,E1,23,10,F7,C1,04
1140 DATA 3A,40,A1,3C,B8,20,D0,C9
1150 REM
1160 REM *** BLOC 6 ***
1170 REM
1180 DATA 00,00,00,21,4F,A1,7E,B7
1190 DATA C8,CD,5A,BB,23,18,F7,0A
1200 DATA 0D,49,4C,20,46,41,55,54
1210 DATA 20,33,20,50,41,52,41,4D
1220 DATA 45,54,52,45,53,0A,0D,00
1230 DATA 21,6D,A1,18,D9,0A,0D,45
1240 DATA 52,52,45,55,52,20,44,45
1250 DATA 20,50,4F,53,49,54,49,4F
1260 REM
1270 REM *** BLOC 7 ***
1280 REM
1290 DATA 4E,0A,0D,00,01,00,BC,0E
1300 DATA 09,ED,49,01,00,BD,3E,00
1310 DATA ED,79,C9,01,00,BC,3E,03
1320 DATA ED,79,01,00,BD,3E,01,ED
1330 DATA 79,C9,00,00,00,00,00,00
1340 REM
1350 REM ***** FIN DE DATA *****
1360 REM
1370 END
```

Après sauvegarde et vérification, vous lancerez ce programme et suivrez les instructions affichées à l'écran.

Pour utilisation, vous entrerez dans votre programme les lignes suivantes :

```
10 MEMORY &9FFF
20 OPENOUT "TRUC."
30 MEMORY HIMEM - 1
40 CLOSEOUT
50 LOAD "EFFETVID.BIN",&A000
60 CALL &A000
```

PROGRAMME DE DÉMONSTRATION

Nous vous conseillons, pour une première utilisation, d'entrer le programme de démonstration suivant, qui permet de s'assurer du bon fonctionnement du programme, surtout si vous avez utilisé le chargeur Basic.

```
10 REM *****
20 REM **
30 REM ** PROGRAMME DE **
40 REM ** DEMONSTRATION DES **
50 REM ** EFFET VIDEO **
60 REM **
70 REM *****
80 REM
90 CALL &BB4E
100 MEMORY &9FFF
110 OPENOUT "BIDON.BID"
120 MEMORY HIMEM - 1
130 CLOSEOUT
140 LOAD "EFFETVID.BIN",&A000
150 CALL &A000
160 MODE 2
170 REM
180 |INIT6845: REM - De preference, tou
jours initialiser le 6845 avant les effe
t videos -
190 REM
200 PHRASE1$ = "Voici une demonstration
d'effet videos"
210 PHRASE2$ = "que vous pourrez reutili
ser pour"
220 PHRASE3$ = "personnaliser vos progra
mmes"
230 PHRASE4$ = "FRAPPEZ SUR UNE TOUCHE P
OUR LA DEMO SUIVANTE"
```

```
240 REM
250 |SCROLLPRINT,10,10,@PHRASE1$
260 |SCROLLPRINT,13,12,@PHRASE2$
270 |SCROLLPRINT,15,14,@PHRASE3$
280 PRINT CHR$(24)
290 |SCROLLPRINT,7,20,@PHRASE4$
300 PRINT CHR$(24)
310 REM
320 CALL &BB06
330 REM
340 |SCROLLPRINT,7,20,"ALLEZ ON ENROULE
POUR LA DEMONSTRATION SUIVANTE"
350 FOR I = 1 TO 1300
360 NEXT I
370 REM
380 |ROULEH
390 REM
400 MODE 1
410 FOR I = 1 TO 12
420 PRINT"VOICI UN ECRAN BIEN REMPLI POU
R ROULER"
430 PRINT
440 NEXT I
450 REM
460 CALL &BB06
470 REM
480 REM
490 |DEROULEH
500 CALL &BB06
510 REM
520 MODE 1
530 PRINT:PRINT:PRINT
540 PRINT"Je suis un ordinateur tres emo
tif"
550 PRINT
560 PRINT"aussi, si vous appuyez mainten
ant"
570 PRINT
580 PRINT"sur une touche, j'aurai peur"
590 REM
600 CALL &BB06
610 REM
620 |TREMBLE
630 REM
640 CALL &BB06
650 REM
660 |INIT6845
670 REM
680 MODE 1
690 PRINT:PRINT:PRINT
```

```
700 PRINT"Une petite ondulation vous pla
irait"
710 PRINT:PRINT:PRINT
720 PRINT"APPUYEZ SUR UNE TOUCHE POUR VO
IR"
730 REM
740 CALL &BB06
750 REM
760 |ONDULE
770 REM
780 CALL &BB06
790 REM
800 |INIT6845
810 REM
820 MODE 0
830 PRINT:PRINT:PRINT
840 |SCROLLPRINT,5,5,"DESTROY !"
850 CALL &BB06
860 |DESTROY
870 CALL &BB06
880 |INIT6845
890 MODE 0
900 PRINT:PRINT:PRINT
910 PRINT "TEXTE ECRASE"
920 |ECRASE
930 CALL &BB06
940 |INIT6845
```

Après lancement, vous verrez apparaître tous les effets décrits plus haut.